
Statistex helps you do common statistics calculations and to explore a data set. It focusses on two things:
	providing you a statistics/2 function that just computes all statistics it knows for a data set, reusing previously made calculations to not compute something again (for instance standard deviation needs the average, so it first computes the average and then passes it on): Statistex.statistics(samples)
	gives you the opportunity to pass known values to functions so that it doesn't need to compute more than it absolutely needs to: Statistex.standard_deviation(samples, average: computed_average)

Installation
def deps do
  [
    {:statistex, "~> 1.0"}
  ]
end
Supported elixir versions are 1.6+ (together with their respective erlang OTP versions aka 19+).
Tests are only running against elixir 1.12+ though, as some dependencies aren't compatible with versions that old.
But also, most people probably don't care about them.
Usage
Check out the documentation of the main Statistex module but here is a small overview:
iex> samples = [1, 3.0, 2.35, 11.0, 1.37, 35, 5.5, 10, 0, 2.35]
# calculate all available statistics at once, efficiently reusing already calculated values
iex> Statistex.statistics(samples)
%Statistex{
  total: 71.57,
  average: 7.156999999999999,
  variance: 109.66060111111112,
  standard_deviation: 10.471895774457991,
  standard_deviation_ratio: 1.4631683351205802,
  median: 2.675,
  percentiles: %{25 => 1.2775, 50 => 2.675, 75 => 10.25},
  frequency_distribution: %{
    0 => 1,
    1 => 1,
    10 => 1,
    35 => 1,
    1.37 => 1,
    2.35 => 2,
    3.0 => 1,
    5.5 => 1,
    11.0 => 1
  },
  mode: 2.35,
  minimum: 0,
  maximum: 35,
  lower_outlier_bound: -12.18125,
  upper_outlier_bound: 23.708750000000002,
  outliers: [35],
  sample_size: 10
}
# or just calculate the value you need
iex> Statistex.average(samples)
7.156999999999999
# Calculate the value you want reusing values you already know
# (check the docs for what functions accepts what options)
iex> Statistex.average(samples, sample_size: 10)
7.156999999999999
# Most Statistex functions raise given an empty list as most functions don't make sense then.
# It is recommended that you manually handle the empty list case should that occur as your
# output is likely also very different from when you have statistics.
iex> Statistex.statistics([])
** (ArgumentError) Passed an empty list ([]) to calculate statistics from, please pass a list containing at least one number.
# You can exclude outliers from the calculation of statistics
iex> Statistex.statistics(samples, exclude_outliers: true)
%Statistex{
  total: 36.57,
  average: 4.0633333333333335,
  variance: 15.696975,
  standard_deviation: 3.96194081227875,
  standard_deviation_ratio: 0.9750469595435808,
  median: 2.35,
  percentiles: %{25 => 1.185, 50 => 2.35, 75 => 7.75},
  frequency_distribution: %{
    0 => 1,
    1 => 1,
    10 => 1,
    1.37 => 1,
    2.35 => 2,
    3.0 => 1,
    5.5 => 1,
    11.0 => 1
  },
  mode: 2.35,
  minimum: 0,
  maximum: 11.0,
  lower_outlier_bound: -12.18125,
  upper_outlier_bound: 23.708750000000002,
  outliers: [35],
  sample_size: 9
}
Supported Statistics
For an up to date overview with explanations please check out the documentation of the Statistex module.
Statistics currently supported:
	average
	frequency_distribution
	maximum
	median
	minimum
	mode
	percentiles
	sample_size
	standard_deviation
	standard_deviation_ratio
	total
	variance
	outliers (determining outliers)
	outlier bounds (lower and upper bound of when samples start being considered outliers)

Alternatives
In elixir there are 2 notable other libraries that I'm aware of: statistics and Numerix.
Both include more functions than just for statistics: general math and more (drawing of random values for instance). They also have more statistics related functions as of this writing. So if you'e looking for something, that Statistex doesn't provide (yet) these are some of the first places I'd look.
Why would you still want to use Statistex?
	statistics/2 is really nice when you're just exploring a data set or just want to have everything at once
	when calling statistics/2 Statistex reuses previously calculated values (average for standard_deviation for instance, or a sorted list of samples for some calculations) which makes for more efficient calculations. Statistex extends that capability to you so that you can pass pre calculated values as optional arguments.
	small and focussed on just statistics :)

We're naturally also looking to add more statistical functions as we go along, and pull requests are very welcome :)
Performance
Statistex is written in pure elixir. C-extensions and friends would surely be faster. The goal of statistex is to be as fast possible in pure elixir while providing correct results. Hence, the focus on reusing previously calculated values and providing that ability to users.
History
Statistex was extracted from benchee and as such it powers benchees statistics calculations. Its great ancestor (if you will) was first conceived in this commit.
Contributing
Contributions to benchee are very welcome! Bug reports, documentation, spelling corrections, new statistics, bugfixes... all of those (and probably more) are much appreciated contributions!
Please respect the Code of Conduct.
You can also look directly at the open issues.
A couple of (hopefully) helpful points:
	Feel free to ask for help and guidance on an issue/PR ("How can I implement this?", "How could I test this?", ...)
	Feel free to open early/not yet complete pull requests to get some early feedback
	When in doubt if something is a good idea open an issue first to discuss it
	In case I don't respond feel free to bump the issue/PR or ping me in other places

Development
	mix deps.get to install dependencies
	mix test to run tests
	mix dialyzer to run dialyzer for type checking, might take a while on the first invocation (try building plts first with mix dialyzer --plt)
	mix credo to find code style problems

Calculate all the statistics for given samples.
Works all at once with statistics/1 or has a lot of functions that can be triggered individually.
To avoid wasting computation, function can be given values they depend on as optional keyword arguments so that these values can be used instead of recalculating them. For an example see average/2.
Most statistics don't really make sense when there are no samples, for that reason all functions except for sample_size/1 raise ArgumentError when handed an empty list.
It is suggested that if it's possible for your program to throw an empty list at Statistex to handle that before handing it to Staistex to take care of the "no reasonable statistics" path entirely separately.
Limitations of ther erlang standard library apply (particularly :math.pow/2 raises for VERY large numbers).

        The optional configuration handed to a lot of functions.



    


    
      
        mode()

      


        Careful with the mode, might be multiple values, one value or nothing.😱 See mode/1.



    


    
      
        percentiles()

      


        The percentiles map returned by percentiles/2.



    


    
      
        sample()

      


        A single sample/



    


    
      
        samples()

      


        The samples to compute statistics from.



    


    
      
        t()

      


        All the statistics statistics/1 computes from the samples.

        average(samples, options \\ [])

      


        Calculate the average.



    


    
      
        frequency_distribution(samples)

      


        A map showing which sample occurs how often in the samples.



    


    
      
        maximum(samples)

      


        The biggest sample.



    


    
      
        median(samples, options \\ [])

      


        Calculates the median of the given samples.



    


    
      
        minimum(samples)

      


        The smallest sample.



    


    
      
        mode(samples, opts \\ [])

      


        Calculates the mode of the given samples.



    


    
      
        outlier_bounds(samples, options \\ [])

      


        Calculates the lower and upper bound for outliers.



    


    
      
        outliers(samples, options \\ [])

      


        Returns all outliers for the given samples, along with the remaining values.



    


    
      
        percentiles(samples, percentiles)

      


        See percentiles/3.



    


    
      
        percentiles(samples, percentiles, options)

      


        Calculates the value at the percentile_rank-th percentile.



    


    
      
        sample_size(samples)

      


        Number of samples in the given list.



    


    
      
        standard_deviation(samples, options \\ [])

      


        Calculate the standard deviation.



    


    
      
        standard_deviation_ratio(samples, options \\ [])

      


          Calculate the standard deviation relative to the average.



    


    
      
        statistics(samples, configuration \\ [])

      


        Calculate all statistics Statistex offers for a given list of numbers.



    


    
      
        total(samples)

      


        The total of all samples added together.



    


    
      
        variance(samples, options \\ [])

      


        Calculate the variance.

          @type configuration() :: keyword()


      


The optional configuration handed to a lot of functions.
Keys used are function dependent and are documented there.

          @type mode() :: [sample()] | sample() | nil


      


Careful with the mode, might be multiple values, one value or nothing.😱 See mode/1.

          @type percentiles() :: %{required(number()) => float()}


      


The percentiles map returned by percentiles/2.

          @type sample() :: number()


      


A single sample/

          @type samples() :: [sample(), ...]


      


The samples to compute statistics from.
Importantly this list is not empty/includes at least one sample otherwise an ArgumentError will be raised.

          @type t() :: %Statistex{
  average: float(),
  frequency_distribution: %{required(sample()) => pos_integer()},
  lower_outlier_bound: number(),
  maximum: number(),
  median: number(),
  minimum: number(),
  mode: mode(),
  outliers: [number()],
  percentiles: percentiles(),
  sample_size: non_neg_integer(),
  standard_deviation: float(),
  standard_deviation_ratio: float(),
  total: number(),
  upper_outlier_bound: number(),
  variance: float()
}


      


All the statistics statistics/1 computes from the samples.
For a description of what a given value means please check out the function here by the same name, it will have an explanation.

          @spec average(
  samples(),
  keyword()
) :: float()


      


Calculate the average.
It's.. well the average.
When the given samples are empty there is no average.
Argumenterror is raised if the given list is empty.
Options
If you already have these values, you can provide both :total and :sample_size. Should you provide both the provided samples are wholly ignored.
Examples
iex> Statistex.average([5])
5.0

iex> Statistex.average([600, 470, 170, 430, 300])
394.0

iex> Statistex.average([-1, 1])
0.0

iex> Statistex.average([2, 3, 4], sample_size: 3)
3.0

iex> Statistex.average([20, 20, 20, 20, 20], total: 100, sample_size: 5)
20.0

iex> Statistex.average(:ignored, total: 100, sample_size: 5)
20.0

iex> Statistex.average([])
** (ArgumentError) Passed an empty list ([]) to calculate statistics from, please pass a list containing at least one number.

          @spec frequency_distribution(samples()) :: %{required(sample()) => pos_integer()}


      


A map showing which sample occurs how often in the samples.
Goes from a concrete occurence of the sample to the number of times it was observed in the samples.
Argumenterror is raised if the given list is empty.
Examples
iex> Statistex.frequency_distribution([1, 2, 4.23, 7, 2, 99])
%{
  2 => 2,
  1 => 1,
  4.23 => 1,
  7 => 1,
  99 => 1
}

iex> Statistex.frequency_distribution([])
** (ArgumentError) Passed an empty list ([]) to calculate statistics from, please pass a list containing at least one number.

          @spec maximum(samples()) :: sample()


      


The biggest sample.
Argumenterror is raised if the given list is empty.
Examples
iex> Statistex.maximum([1, 100, 24])
100

iex> Statistex.maximum([])
** (ArgumentError) Passed an empty list ([]) to calculate statistics from, please pass a list containing at least one number.

          @spec median(
  samples(),
  keyword()
) :: number()


      


Calculates the median of the given samples.
The median can be thought of separating the higher half from the lower half of the samples.
When all samples are sorted, this is the middle value (or average of the two middle values when the number of times is even).
More stable than the average.
Argumenterror is raised if the given list is empty.
Options
	:percentiles - you can pass it a map of calculated percentiles to fetch the median from (it is the 50th percentile).
If it doesn't include the median/50th percentile - it will still be computed.
	:sorted?: indicating the samples you're passing in are already sorted. Defaults to false. Only set this,
if they are truly sorted - otherwise your results will be wrong. Sorting only occurs when percentiles aren't provided.

Examples
iex> Statistex.median([1, 3, 4, 6, 7, 8, 9])
6.0

iex> Statistex.median([1, 3, 4, 6, 7, 8, 9], percentiles: %{50 => 6.0})
6.0

iex> Statistex.median([1, 3, 4, 6, 7, 8, 9], percentiles: %{25 => 3.0})
6.0

iex> Statistex.median([1, 3, 4, 6, 7, 8, 9], sorted?: true)
6.0

iex> Statistex.median([1, 2, 3, 4, 5, 6, 8, 9])
4.5

iex> Statistex.median([0])
0.0

iex> Statistex.median([])
** (ArgumentError) Passed an empty list ([]) to calculate statistics from, please pass a list containing at least one number.

          @spec minimum(samples()) :: sample()


      


The smallest sample.
Argumenterror is raised if the given list is empty.
Examples
iex> Statistex.minimum([1, 100, 24])
1

iex> Statistex.minimum([])
** (ArgumentError) Passed an empty list ([]) to calculate statistics from, please pass a list containing at least one number.

          @spec mode(
  samples(),
  keyword()
) :: mode()


      


Calculates the mode of the given samples.
Mode is the sample(s) that occur the most. Often one value, but can be multiple values if they occur the same amount of times. If no value occurs at least twice, there is no mode and it hence returns nil.
Argumenterror is raised if the given list is empty.
Options
If already calculated, the :frequency_distribution option can be provided to avoid recalulating it.
Examples
iex> Statistex.mode([5, 3, 4, 5, 1, 3, 1, 3])
3

iex> Statistex.mode([1, 2, 3, 4, 5])
nil

# When a measurement failed and nils is reported as the only value
iex> Statistex.mode([nil])
nil

iex> Statistex.mode([])
** (ArgumentError) Passed an empty list ([]) to calculate statistics from, please pass a list containing at least one number.

iex> mode = Statistex.mode([5, 3, 4, 5, 1, 3, 1])
iex> Enum.sort(mode)
[1, 3, 5]

          @spec outlier_bounds(
  samples(),
  keyword()
) :: {lower :: number(), upper :: number()}


      


Calculates the lower and upper bound for outliers.
Any sample that is < as the lower bound and any sample > are outliers of
the given samples.
List passed needs to be non empty, otherwise an ArgumentError is raised.
Options
	:percentiles - you can pass it a map of calculated percentiles (25th and 75th are needed).
If it doesn't include them - it will still be computed.
	:sorted?: indicating the samples you're passing in are already sorted. Defaults to false. Only set this,
if they are truly sorted - otherwise your results will be wrong. Sorting only occurs when percentiles aren't provided.

Examples
iex> Statistex.outlier_bounds([3, 4, 5])
{0.0, 8.0}

iex> Statistex.outlier_bounds([4, 5, 3])
{0.0, 8.0}

iex> Statistex.outlier_bounds([3, 4, 5], sorted?: true)
{0.0, 8.0}

iex> Statistex.outlier_bounds([3, 4, 5], percentiles: %{25 => 3.0, 75 => 5.0})
{0.0, 8.0}

iex> Statistex.outlier_bounds([1, 2, 6, 50, 50, 50, 50, 50, 50, 50, 50, 50, 50, 50])
{22.5, 66.5}

iex> Statistex.outlier_bounds([50, 50, 50, 50, 50, 50, 50, 50, 50, 50, 50, 99, 99, 99])
{31.625, 80.625}

iex> Statistex.outlier_bounds([])
** (ArgumentError) Passed an empty list ([]) to calculate statistics from, please pass a list containing at least one number.

          @spec outliers(
  samples(),
  keyword()
) :: {samples() | [], samples()}


      


Returns all outliers for the given samples, along with the remaining values.
Returns: {outliers, remaining_samples} where remaining_samples has the outliers removed.
Argumenterror is raised if the given list is empty.
Options
	:outlier_bounds - if you already have calculated the outlier bounds.
	:percentiles - you can pass it a map of calculated percentiles (25th and 75th are needed).
If it doesn't include them - it will still be computed.
	:sorted?: indicating the samples you're passing in are already sorted. Defaults to false. Only set this,
if they are truly sorted - otherwise your results will be wrong. Sorting only occurs when percentiles aren't provided.

Examples
iex> Statistex.outliers([3, 4, 5])
{[], [3, 4, 5]}

iex> Statistex.outliers([1, 2, 6, 50, 50, 50, 50, 50, 50, 50, 50, 50, 50, 50])
{[1, 2, 6], [50, 50, 50, 50, 50, 50, 50, 50, 50, 50, 50]}

iex> Statistex.outliers([50, 50, 1, 50, 50, 50, 50, 50, 2, 50, 50, 50, 50, 6])
{[1, 2, 6], [50, 50, 50, 50, 50, 50, 50, 50, 50, 50, 50]}

iex> Statistex.outliers([50, 50, 50, 50, 50, 50, 50, 50, 50, 50, 50, 99, 99, 99])
{[99, 99, 99], [50, 50, 50, 50, 50, 50, 50, 50, 50, 50, 50]}

iex> Statistex.outliers([])
** (ArgumentError) Passed an empty list ([]) to calculate statistics from, please pass a list containing at least one number.

          @spec percentiles(samples(), number() | [number(), ...]) :: percentiles()


      


See percentiles/3.

  



  
    
      
    
    
      percentiles(samples, percentiles, options)



        
          
        

    

  


  

Calculates the value at the percentile_rank-th percentile.
Think of this as the value below which percentile_rank percent of the samples lie.
For example, if Statistex.percentiles(samples, 99) == 123.45,
99% of samples are less than 123.45.
Passing a number for percentile_rank calculates a single percentile.
Passing a list of numbers calculates multiple percentiles, and returns them
as a map like %{90 => 45.6, 99 => 78.9}, where the keys are the percentile
numbers, and the values are the percentile values.
Percentiles must be between 0 and 100 (excluding the boundaries).
The method used for interpolation is described here and recommended by NIST.
Argumenterror is raised if the given list is empty.
Options
	:sorted?: indicating the samples you're passing in are already sorted. Defaults to false. Only set this,
if they are truly sorted - otherwise your results will be wrong.

Examples
iex> Statistex.percentiles([5, 3, 4, 5, 1, 3, 1, 3], 12.5)
%{12.5 => 1.0}

iex> Statistex.percentiles([1, 1, 3, 3, 3, 4, 5, 5], 12.5, sorted?: true)
%{12.5 => 1.0}

iex> Statistex.percentiles([5, 3, 4, 5, 1, 3, 1, 3], [50])
%{50 => 3.0}

iex> Statistex.percentiles([5, 3, 4, 5, 1, 3, 1, 3], [75])
%{75 => 4.75}

iex> Statistex.percentiles([5, 3, 4, 5, 1, 3, 1, 3], 99)
%{99 => 5.0}

iex> Statistex.percentiles([5, 3, 4, 5, 1, 3, 1, 3], [50, 75, 99])
%{50 => 3.0, 75 => 4.75, 99 => 5.0}

iex> Statistex.percentiles([5, 3, 4, 5, 1, 3, 1, 3], 100)
** (ArgumentError) percentile must be between 0 and 100, got: 100

iex> Statistex.percentiles([5, 3, 4, 5, 1, 3, 1, 3], 0)
** (ArgumentError) percentile must be between 0 and 100, got: 0

iex> Statistex.percentiles([], [50])
** (ArgumentError) Passed an empty list ([]) to calculate statistics from, please pass a list containing at least one number.

          @spec sample_size([sample()]) :: non_neg_integer()


      


Number of samples in the given list.
Nothing to fancy here, this just calls length(list) and is only provided for completeness sake.
Examples
iex> Statistex.sample_size([])
0

iex> Statistex.sample_size([1, 1, 1, 1, 1])
5

          @spec standard_deviation(
  samples(),
  keyword()
) :: float()


      


Calculate the standard deviation.
A measurement how much samples vary (the higher the more the samples vary). It's the square root of the variance. Unlike the variance, its unit is the same as that of the sample (as calculating the variance includes squaring).
Options
If already calculated, the :variance option can be provided to avoid recalulating those values.
Argumenterror is raised if the given list is empty.
Examples
iex> Statistex.standard_deviation([4, 9, 11, 12, 17, 5, 8, 12, 12])
4.0

iex> Statistex.standard_deviation(:dontcare, variance: 16.0)
4.0

iex> Statistex.standard_deviation([42])
0.0

iex> Statistex.standard_deviation([1, 1, 1, 1, 1, 1, 1])
0.0

iex> Statistex.standard_deviation([])
** (ArgumentError) Passed an empty list ([]) to calculate statistics from, please pass a list containing at least one number.

          @spec standard_deviation_ratio(
  samples(),
  keyword()
) :: float()


      


  Calculate the standard deviation relative to the average.
  This helps put the absolute standard deviation value into perspective expressing it relative to the average. It's what percentage of the absolute value of the average the variance takes.
  Argumenterror is raised if the given list is empty.
  ## Options
  If already calculated, the :average and :standard_deviation options can be provided to avoid recalulating those values.
  If both values are provided, the provided samples will be ignored.
  ## Examples
  iex> Statistex.standard_deviation_ratio([4, 9, 11, 12, 17, 5, 8, 12, 12])
  0.4

  iex> Statistex.standard_deviation_ratio([-4, -9, -11, -12, -17, -5, -8, -12, -12])
  0.4

  iex> Statistex.standard_deviation_ratio([4, 9, 11, 12, 17, 5, 8, 12, 12], average: 10.0, standard_deviation: 4.0)
  0.4

  iex> Statistex.standard_deviation_ratio(:ignored, average: 10.0, standard_deviation: 4.0)
  0.4

  iex> Statistex.standard_deviation_ratio([])
  ** (ArgumentError) Passed an empty list ([]) to calculate statistics from, please pass a list containing at least one number.

          @spec statistics(samples(), configuration()) :: t()


      


Calculate all statistics Statistex offers for a given list of numbers.
The statistics themselves are described in the individual samples that can be used to calculate individual values.
ArgumentError is raised if the given list is empty.
Options
	:percentiles: percentiles to calculate (see percentiles/2).
The percentiles 25th, 50th (median) and 75th are always calculated.
	:exclude_outliers can be set to true or false. Defaults to false.
If this option is set to true the outliers are excluded from the calculation
of the statistics.
	:sorted?: indicating the samples you're passing in are already sorted. Defaults to false. Only set this,
if they are truly sorted - otherwise your results will be wrong.

Examples
iex> Statistex.statistics([50, 50, 450, 450, 450, 500, 500, 500, 600, 900])
%Statistex{
  total: 4450,
  average: 445.0,
  variance: 61_361.11111111111,
  standard_deviation: 247.71175004652304,
  standard_deviation_ratio: 0.5566556180820742,
  median: 475.0,
  percentiles: %{25 => 350.0, 50 => 475.0, 75 => 525.0},
  frequency_distribution: %{50 => 2, 450 => 3, 500 => 3, 600 => 1, 900 => 1},
  mode: [500, 450],
  minimum: 50,
  maximum: 900,
  lower_outlier_bound: 87.5,
  upper_outlier_bound: 787.5,
  outliers: [50, 50, 900],
  sample_size: 10
}

# excluding outliers changes the results
iex> Statistex.statistics([50, 50, 450, 450, 450, 500, 500, 500, 600, 900], exclude_outliers: true)
%Statistex{
  total: 3450,
  average: 492.85714285714283,
  variance: 2857.142857142857,
  standard_deviation: 53.452248382484875,
  standard_deviation_ratio: 0.1084538372977954,
  median: 500.0,
  percentiles: %{25 => 450.0, 50 => 500.0, 75 => 500.0},
  frequency_distribution: %{450 => 3, 500 => 3, 600 => 1},
  mode: [500, 450],
  maximum: 600,
  minimum: 450,
  lower_outlier_bound: 87.5,
  upper_outlier_bound: 787.5,
  outliers: [50, 50, 900],
  sample_size: 7
}

iex> Statistex.statistics([])
** (ArgumentError) Passed an empty list ([]) to calculate statistics from, please pass a list containing at least one number.

          @spec total(samples()) :: number()


      


The total of all samples added together.
Argumenterror is raised if the given list is empty.
Examples
iex> Statistex.total([1, 2, 3, 4, 5])
15

iex> Statistex.total([10, 10.5, 5])
25.5

iex> Statistex.total([-10, 5, 3, 2])
0

iex> Statistex.total([])
** (ArgumentError) Passed an empty list ([]) to calculate statistics from, please pass a list containing at least one number.

          @spec variance(
  samples(),
  keyword()
) :: float()


      


Calculate the variance.
A measurement how much samples vary (the higher the more the samples vary). This is the variance of a sample and is hence in its calculation divided by sample_size - 1 (Bessel's correction).
Argumenterror is raised if the given list is empty.
Options
If already calculated, the :average and :sample_size options can be provided to avoid recalulating those values.
Examples
iex> Statistex.variance([4, 9, 11, 12, 17, 5, 8, 12, 12])
16.0

iex> Statistex.variance([4, 9, 11, 12, 17, 5, 8, 12, 12], sample_size: 9, average: 10.0)
16.0

iex> Statistex.variance([42])
0.0

iex> Statistex.variance([1, 1, 1, 1, 1, 1, 1])
0.0

iex> Statistex.variance([])
** (ArgumentError) Passed an empty list ([]) to calculate statistics from, please pass a list containing at least one number.

